Pensieve Header: Figure out if there is a homomorphic expansion for red-over-green tangles.

```
Basis[n_] := DeleteCases[
   Permutations[P @@ Range[n]],
   Alternatives[
    P[___, i_, j_, ___] /; i == j + 1,
    P[___, i_, j_, k_, ___] /; i == j + 2 && k == j + 1
   ]
  ];
P /: p1_P ** p2_P := Join[p1, (Length[p1] + #) & /@ p2];
ASeries /: c_?NumberQ * a_ASeries := Expand[c #] & /@ a;
ASeries /: a1_ASeries + a2_ASeries := Module[
   {m = Min[Length[a1], Length[a2]]},
   ASeries @@ (Take[List @@ a1, m] + Take[List @@ a2, m])
  ];
ASeries /: a1_ASeries ** a2_ASeries := Module[
   {m = Min[Length[a1], Length[a2]] - 1, P1, P2},
   ASeries @@ Table[
      Sum[
       Expand[(a1[[d1 + 1]] /. P → P1) * (a2[[d - d1 + 1]] /. P → P2)],
       {d1, 0, d}
      ],
      {d, 0, m}
     ] /. p1_P1 * p2_P2 :→ (P @@ p1) ** (P @@ p2)
  ];
Invert[ASeries[P[], x___]] := Module[
   {s, t, k},
   t = ReplacePart[ASeries @@ Table[0, {1 + Length[{x}]}], 1 → P[]];
   t + Sum[
     t = (-t) ** ASeries[0, x],
     {Length[{x}]}
    ]
  ];
GT[p_P] := Reverse[p] //. {
    P[l___, i_, j_, r___] /; i == j + 1 :→ P[l, j, i, r],
    P[l___, i_, j_, k_, r___] /; i == j + 2 && k == j + 1 :→ P[l, j, k, i, r]
   };
GT[expr_] := expr /. p_P :→ GT[p];
EQ1[R_ASeries] := R ** Invert[R];
EQ2[R_ASeries] := GT[R] ** GT[Invert[R]];
EQ[R_ASeries] := {EQ1[R], EQ2[R]};
```

```
Basis /@ Range[0, 4]
```

```
{{P[]}, {P[1]}, {P[1, 2]}, {P[1, 2, 3], P[2, 3, 1]}, {P[1, 2, 3, 4], P[1, 3, 4, 2],
  P[2, 3, 1, 4], P[2, 3, 4, 1], P[2, 4, 1, 3], P[3, 1, 4, 2], P[3, 4, 1, 2], P[4, 1, 2, 3]}}
```

```
GT[Basis /@ Range[0, 4]]
```

```
{{P[]}, {P[1]}, {P[1, 2]}, {P[2, 3, 1], P[1, 2, 3]}, {P[3, 4, 1, 2], P[2, 3, 4, 1],
  P[4, 1, 2, 3], P[1, 3, 4, 2], P[3, 1, 4, 2], P[2, 4, 1, 3], P[1, 2, 3, 4], P[2, 3, 1, 4]}}
```

```
R2 = ASeries[P[], P[1], a2 P[1, 2]]
```

```
ASeries[P[], P[1], a2 P[1, 2]]
```

```
Invert[R2]

ASeries[P[], -P[1], P[1, 2] - a2 P[1, 2]]

EQ[R2]

{ASeries[P[], 0, 0], ASeries[P[], 0, 0]}

R3p = Append[R2, Array[a3, {Length[Basis[3]]}].Basis[3]]

ASeries[P[], P[1], a2 P[1, 2], a3[1] P[1, 2, 3] + a3[2] P[2, 3, 1]]

Invert[R3p]

ASeries[P[], -P[1], P[1, 2] - a2 P[1, 2],
 -P[1, 2, 3] + 2 a2 P[1, 2, 3] - a3[1] P[1, 2, 3] - a3[2] P[2, 3, 1]]

EQ[R3p]

{ASeries[P[], 0, 0, 0],
 ASeries[P[], 0, 0, P[1, 2, 3] - 2 a2 P[1, 2, 3] - P[2, 3, 1] + 2 a2 P[2, 3, 1]]}

R3 = R3p /. {a2 → 1 / 2}
```

$$\text{ASeries}\left[P[], P[1], \frac{1}{2} P[1, 2], a3[1] P[1, 2, 3] + a3[2] P[2, 3, 1]\right]$$

```
EQ[R3]

{ASeries[P[], 0, 0, 0], ASeries[P[], 0, 0, 0]}

R4p = Append[R3, Array[a4, {Length[Basis[4]]}].Basis[4]]
```

$$\text{ASeries}\Big[P[], P[1], \frac{1}{2} P[1, 2], a3[1] P[1, 2, 3] + a3[2] P[2, 3, 1], a4[1] P[1, 2, 3, 4] + a4[2] P[1, 3, 4, 2] + a4[3] P[2, 3, 1, 4] + a4[4] P[2, 3, 4, 1] + a4[5] P[2, 4, 1, 3] + a4[6] P[3, 1, 4, 2] + a4[7] P[3, 4, 1, 2] + a4[8] P[4, 1, 2, 3]\Big]$$

```
EQ[R4p]
```

$$\Big\{\text{ASeries}[P[], 0, 0, 0, 0], \text{ASeries}\Big[P[], 0, 0, 0, \frac{1}{4} P[1, 2, 3, 4] - 2\, a3[2] P[1, 2, 3, 4] - a3[1] P[1, 3, 4, 2] - a3[1] P[2, 3, 1, 4] + a3[2] P[2, 3, 4, 1] - \frac{1}{4} P[3, 4, 1, 2] + 2\, a3[1] P[3, 4, 1, 2] + a3[2] P[4, 1, 2, 3]\Big]\Big\}$$

```
{R4 = R4p /. {a3[1] → 0, a3[2] → 1 / 8}, EQ[R4]}
```

$$\Big\{\text{ASeries}\Big[P[], P[1], \frac{1}{2} P[1, 2], \frac{1}{8} P[2, 3, 1], a4[1] P[1, 2, 3, 4] + a4[2] P[1, 3, 4, 2] + a4[3] P[2, 3, 1, 4] + a4[4] P[2, 3, 4, 1] + a4[5] P[2, 4, 1, 3] + a4[6] P[3, 1, 4, 2] + a4[7] P[3, 4, 1, 2] + a4[8] P[4, 1, 2, 3]\Big], \Big\{\text{ASeries}[P[], 0, 0, 0, 0], \text{ASeries}\Big[P[], 0, 0, 0, \frac{1}{8} P[2, 3, 4, 1] - \frac{1}{4} P[3, 4, 1, 2] + \frac{1}{8} P[4, 1, 2, 3]\Big]\Big\}\Big\}$$